


Julio Le Parc et art optique

L'Art optique rassemble des artistes fascinés par ce que l'œil perçoit du mouvement et de la lumière et qui en explorent les propriétés : intensité, rythme, variabilité, cycles... La démarche est bien sûr esthétique (non technique, physique ou métaphysique). C'est un mouvement riche et varié, qui s'exprime par des tableaux, des sculptures mais aussi des assemblages, installations ou montages vidéos. À ses débuts, on parle d'Art cinétique et optique ; dans les années 1960, l'appellation est simplifiée : on parle d'Art optique en France, d'Optique Art ou Op Art chez les anglo-saxons.

Julio Le Parc est né en 1928 à Mendoza en Argentine. Il vit et travaille à Cachan. Précurseur de l'art cinétique et de l'Op Art, il est membre fondateur du G.R.A.V. (Groupe de Recherche d'Art Visuel) avec Horacio Garcia Rossi, François Morellet, Francisco Sobrino, Joël Stein et Jean-Pierre Vasarely. Il travaille sur de multiples support : peintures, sculptures, installations lumineuses. Julio Le Parc est un personnage emblématique de l'histoire de l'art. Artiste engagé, il fut expulsé de France en mai 1968 pour sa participation à l'atelier populaire et ses manifestations contre les institutions. Défenseur des droits de l'homme, il lutta contre les dictatures d'Amérique Latine. Personnalité entière, il refusa en 1972 une rétrospective au musée d'Art moderne de la Ville de Paris en la jouant à pile ou face. (d'après la biographie de Julio Le Parc présente sur le site du palais de Tokyo à Paris)

Étude géométrique de l'oeuvre

Le motif de base est un portion de disque (fig 1). Ce motif est répété un certain nombre de fois et disposé en grille pour composer le tableau. L'ensemble peut-être décomposé en 4 morceaux (fig 2). Pour la suite, l'étude portera sur la partie en haut à gauche (fig 3). Les autres parties se déduisent de la première par des rotations successives de 90°.




L'orientation des motifs est déterminée par une rotation autour de leur centre que l'on choisit parmi les treize positions suivantes. Ces treize positions correspondent à la première ligne de la fig 3 :



L'angle unitaire de rotation est donc donné par : $180 \div 12 = 15$

Les angles de rotation du motif de base sont donc : 0° ; 15° ; 30° ; 45° ; 60° ; 75° ; 90° ; 105° ; 120° ; 135° ; 150° ; 165° ; 180°.

Les autres lignes de la figure sont construites de la même manière que la première avec une rotation supplémentaire de 15° de tous les motifs. Il faut cumuler cette rotation à chaque changement de ligne.



Éléments de réponse avec *GéoTortue*

Le rayon choisi pour l'arc de cercle est 50. Il faut que la portion de disque soit centrée, c'est à dire que la tortue se situe au centre du cercle. De plus, il faut donner la couleur au motif par programmation. Il n'est pas concevable de remplir avec l'interface de coloriage les 676 arcs de cercles. Pour réaliser un cercle par programmation, nous utilisons un polygone à 360 côtés.

cercle approché par un polygone à 360 côtés :


```

pour cercle rayon
td 0.5
rep 360 (av 2*π*rayon/360 ; td 1)
tg 0.5
fin
    
```

arcs de cercle correspondant :

```

pour arccercle rayon angle
td 0.5
rep angle (av 2*π*rayon/360;td 1)
tg 0.5
fin
    
```



```

pour motif
tg 45 ; av 50 ; td 90
remplis arccercle 50 270
td 90 ; av 50 ; tg 45
fin
    
```

Pour disposer les motifs dans une grille, nous pouvons créer une procédure ligne afin de clarifier le code. Les motifs appartenant à une ligne sont disposés tous les 120 pas de tortue. Sans appliquer de rotations, la première ligne donne :



```

pour ligne1
rep 13 (motif ; lc ; td 90 ; av 120 ; tg 90 ; bc)
lc ; tg 90 ; av 13*120 ; td 90 ; bc
fin
    
```

Il reste à appliquer une rotation de 15° qui se cumule à chaque motif. Les angles de rotations sont des multiples de 15 ayant pour expression : $n \times 15$ avec $0 \leq n < 12$.

Dans la procédure, il faut donc incrémenter une variable dans la boucle *rep*. La première ligne donne :



```
pour ligne1
num:=0
rep 13 (td (num*15) ; motif ; tg (num*15) ; lc ; td 90 ; av 120 ; tg 90 ; bc ; num:=num+1)
lc ; tg 90 ; av 13*120 ; td 90 ; bc
fin
```

Pour les lignes suivantes, le même principe est appliqué avec une rotation supplémentaire de 15° pour le motif de départ. Ainsi pour la deuxième ligne, les angles seront de la forme : $n \times 15$ avec $1 \leq n < 13$. Nous obtenons pour la deuxième ligne :



```
pour ligne2
num:=1
rep 13 (td (num*15) ; motif ; tg (num*15) ; lc ; td 90 ; av 120 ; tg 90 ; bc ; num:=num+1)
lc ; tg 90 ; av 13*120 ; td 90 ; bc
fin
```

Et ainsi de suite jusqu'à la treizième ligne. Nous pouvons donc remplacer toutes ces procédures par une seule en prenant comme variable le numéro de la ligne :


```
pour ligne numéro
num:=numéro
rep 13 (td (num*15) ; motif ; tg (num*15) ; lc ; td 90 ; av 120 ; tg 90 ; bc ; num:=num+1)
lc ; tg 90 ; av 13*120 ; td 90 ; bc
fin
```

Il nous faut désormais construire la procédure correspondant à la *fig 3* en réalisant une procédure grille en incrémentant le numéro de la ligne :

```
pour grille
m:=0
rep 13 (ligne m ; lc ; re 120 ; bc ; m:=m+1)
lc ; av 13*120 ; bc
fin
```

Il ne nous reste plus qu'à réaliser la figure finale en disposant les quatre grilles correctement :

```
pour instabilité
rep 4 (grille ; lc ; td 90 ; av 25*120 ; bc)
fin
```




Avec les élèves

Ce travail est réalisable en fin de collège à condition que les élèves soient habitués à utiliser *GéoTortue* et qu'ils en maîtrisent les rudiments : utilisation de la commande *rep*, réalisation de pavages ...

L'incrémentation de la boucle *rep* (qui correspond à une boucle *for*) peut poser problème car nous rentrons dans de la programmation algorithmique. Cela en fait une activité tout à fait intéressante en classe de seconde. Il est possible de pallier ce problème en écrivant treize procédures ligne (ligne1, ligne2 ...).

Nous pouvons donner la procédure *arcercle* ou choisir de ne pas le faire. L'étude des procédures *cercle* et *arcercle* peut être réalisée au collège.


Pour les classes en début de collège, cette activité peut se révéler trop ambitieuse. Nous pouvons proposer la figure ci-contre qui reprend le même principe avec beaucoup moins de motifs de base, les angles de 15° étant remplacés par des angles de 45° .



Il est aussi nécessaire d'accompagner cette activité d'un travail de recherche sur l'artiste, sur le mouvement dans lequel il s'inscrit, sur l'œuvre elle-même (sa date de réalisation, ses dimensions, la matière utilisée, ...), d'insister sur la différence entre une reproduction à l'ordinateur et le tableau lui-même.

Pour aller plus loin

D'autres réalisations sont possibles à partir du programme précédent. Il suffit de changer quelques paramètres comme le motif de départ, la dimension de la grille ... Dans l'exemple suivant, le motif de base est un carré.



Sources et liens

- site officiel de Julio Le Parc
- site du palais de Tokyo de Paris
- site du centre Georges Pompidou de Paris
- brochure de l'exposition *dynamo* du Grand Palais (Paris)
- site de l'IREM Paris Nord
- site de *GéoTortue*